

Nauwkeurige D/A-converter in UNIFACE.

Hans Zeedijk


Resolutie in 12 bits.

De DDAC12-print, die in de PTC-winkel verkrijgbaar is, is een fraaie uitbreiding van het pakket printen in Uniface. De print biedt 2 analoge uitgangen, die instelbare spanningsbereiken kunnen hebben in 12 bits resolutie en dit ook werkelijk tot de laatste bit betrouwbaar. Zowel negatieve als positieve bereiken van 5 Volt en 10 Volt zijn te verkrijgen met behulp van jumper-instellingen en de twee kanalen kunnen een verschillend schaalbereik hebben, b.v. 0-+5 Volt voor kanaal 1 en 0- -5 Volt voor kanaal 2. De conversietijd bedraagt slechts 1,5 microseconde, zodat in het algemeen het stuurprogramma de snelheid zal bepalen, afhankelijk van de gebruikte computer.

Mijlpaal in het Uniface-gebeuren.

Met de nieuwe DDAC12-print is een fase afgesloten in de Uniface-ontwikkeling. Er zijn nu naast de bekende digitale printen twee sets analoge printen: een goedkope serie, die vooral voor hobby-toepassingen geschikt is met een resolutie in 8 bits; en een duurere professionele serie voor "het betere werk" met een resolutie in 12 bits. De eerste serie werkt direct op de interface bandkabel gevoed door de computer-voeding; de tweede serie heeft naast de +5 Volt ook de +15 en -15 Volt voeding nodig. Het afgeronde Uniface pakket bestaat nu uit:

- Interfaces voor de P2000, de MSX(1 + 2), en de PC
 - Connecties: bandkabel en bufferprint
 - Functionele printen: digitaal (8 bits in en uit)
 - analoog A/D en D/A in 8 bits
 - analoog A/D en D/A in 12 bits
- Deze mijlpaal is geen eindpunt. Er zijn nieuwe hardware ontwikkelingen in voorbereiding, o.a. een Commodore 64/128-interface, een experimenteerprint en een multiplexer-print voor uitbreiding van het aantal analoge ingangen bij de ADC8. Er komen ook uitbreidingen in de vorm van hulpmiddelen en in de toekomst zal meer dan tot nu aandacht besteed worden aan applicaties.



Afb. 1. Het verband tussen weerstand en temperatuur.

Temperatuurmeting met UNIFACE.

De goedkoopste temperatuur-sensoren voor het gebied van -50 tot +125 °C (prijs vanaf f.1,50) zijn NTC- en PTC-weerstanden. NTC staat voor negative temperature coefficient, d.w.z. de weerstand van de sensor daalt met toenemende temperatuur. PTC-weerstanden zijn niet die lastige mensen in uw afdeling, maar weerstanden die groter worden met toenemende temperatuur. We gaan alleen verder over de NTC's.

Bijgaande grafiek (overgenomen uit de TELECOM-catalogus) geeft het verband tussen de weerstandswaarden en de


temperatuur. Bij benadering geldt: $\log(\text{weerstand}) = B \cdot \text{temperatuur} + \text{constante}$

B is negatief en heeft vaak de waarde van -0,02. In feite is B niet echt constant, maar ook temperatuursafhankelijk. Ook de werkelijke weerstandswaarden van NTC's kloppen zelden: als 10k wordt opgegeven bij 20°C kan de afwijking wel 10 à 20% bedragen. Een uitzondering vormen de NTC's uit de Accuracy Line van Philips, die binnen 1% constant zijn, een vrijwel constante B-waarde bezitten, en bovendien bij gebruik geen drift vertonen.

We zullen nu twee technieken behandelen om met NTC-weerstanden de temperatuur te meten met gebruik van UNIFACE.

- Met een analoge input-print ADC8 of ADC12.

De NTC-weerstand R1 wordt in serie met de niet-temperatuurgevoelige weerstand R2 belast met een spanning, b.v. 5 Volt (op beide ADC-prints aanwezig) volgens bijgaand schema:



De op de analoge ingang gemeten spanning is:

$$V_{in} = 5 \cdot R_2 / (R_1 + R_2) \text{ Volt ofwel } R_1 = 5 \cdot R_2 / V_{in} - R_2$$

De waarde van de NTC-weerstand bedraagt bij 20°C b.v. 100k (in de winkel zijn vele waarden verkrijgbaar). Toepassing van de NTC-weerstand/temperatuur-vergelijking levert:

$$\log(R_1/100) = -0,02 \cdot (T-20)$$

waarin R1 de weerstandswaarde is bij de temperatuur T.

$$\text{Omwerken levert: } T = 20 - 50 \cdot \log(R_1/100)$$

Invullen van de gemeten waarde R1 geeft als resultaat de temperatuur T van de NTC-weerstand.


Het programma DA-TEMP.BAS, waarvan bijgaand de listing, geeft een voorbeeld van de temperatuurmeting, waarbij iedere 0,1 seconde een meetwaarde wordt geproduceerd en in een grafiek op het scherm uitgezet.

Dit programma in GWBASIC is geschikt voor de PC, maar gemakkelijk om te zetten in een P2000- of MSX-BASIC programma. Uitleg van het programma is als commentaar in de listing opgenomen.

- Met een INPUT en DIGOUT-print.

De temperatuurmeting is in wezen een weerstandsmeting en dit is ook uit te voeren door de ontlading van een condensator over de NTC te meten. In Uniface kan dit met behulp van een DIGOUT- en een INPUT-print.

De schakeling geeft aan hoe dit moet. Met de DIGOUT-print wordt de condensator opgeladen tot b.v. 5 Volt.



Hiervoor wordt 1 uitgangsbij gebruik. De spanning over de condensator wordt gemeten met één bit van de INPUT-print, die bij 5 Volt een nullaat uitlezen op deze bit. De meting start door de DIGOUT-outputbit nul te maken. De condensator ontladst over de NTC-weerstand en de ontladstroom is afhankelijk van de waarde van deze weerstand en daardoor ook de tijd die nodig is om de spanning over de condensator te laten dalen tot 2,5 Volt. Op dat punt verandert de bit-waarde van de INPUT-print van nul in één en dat markeert de tijd die voor de ontlading nodig was. Hieruit valt nu de waarde van de NTC-weerstand te bepalen. Immers de ontladstroom bedraagt:

$$i = C \cdot dV/dt = V/R$$

Integratie levert voor de spanning op tijd t:

$$\text{(natuurlijke)} \log(V/V_0) = -t/RC$$

Omslag vindt plaats als $V = 1/2 V_0$; de natuurlijke logaritme bedraagt dan -0,693. Ofwel de tijd van omslag bedraagt: $t = 0,693 \cdot RC$

Als de omslagtijd gemeten is geldt voor R:

$$R = t / (0,693 \cdot RC)$$

Uit de weerstand laat zich op de bekende wijze weer de temperatuur berekenen. Het programma IO-TEMP.BAS print na iedere meting de temperatuur op het scherm (uitleg in de listing). Door C groot te kiezen b.v. 100 microF is de tijd nauwkeurig te meten (bij NTC-weerstand van 100k en 20°C ongeveer 7 seconden).

De meetsnelheid is beperkt, omdat na iedere meting de condensator weer opgeladen moet worden en in de gegeven voorbeeldswaarde duurt dit ongeveer 1 minuut. Met 1 DIGOUT-print en 1 INPUT-print kunnen 8 temperatuursensoren worden aangesloten op de computer.

Vraag over de UNIFACE-INPUT-print.

Door A. Langenakens uit Heerlen is de vraag gesteld hoe precies het aflezen van de stand van een schakelaar met de INPUT-print van het UNIFACE-systeem geprogrammeerd moet worden.

Sluit de schakelaar aan tussen GRND en b.v. bit 4 van de INPUT-print. Zet alle posities van de 8-polige switch op de print op ON met uitzondering van de positie 1. Deze staat dus op OFF. Het adres van de print is nu 254 en in het programma moet hiervan de inverse waarde gebruikt worden, dus 255-254=1.

We kunnen nu controleren of de print goed geadresseerd wordt met de twee BASIC instructies:
 OUT X+1,1:PRINT INP(X+1)

X is het interface-adres, te weten 48 voor een MSX, 96 voor een P2000 en 784 voor een PC. Vul derhalve het getal in op de plaats van X, dat behoort bij de door U gebruikte computer. Het antwoord dat de computer op het scherm print is juist, wanneer het het getal 61 is. 255 betekent, dat de computer het interface niet "ziet" en 63 betekent, dat de print niet juist geadresseerd wordt, waarschijnlijk door een foutief adres op de switch.

Het uitlezen van de input op de INPUT-print gebeurt met:
 OUT X+1,1:PRINT INP(X)

X zijn weer de bekende getallen. Bij open schakelaar antwoordt de computer nu 0 en bij gesloten schakelaar 16 (verbinding met bit 4). Als bit 7 met de GRND verbonden zou zijn, zou het antwoord 128 geweest zijn, en bij bit 2 was het 4 enz.

In feite wordt de spanning van de ingangen bepaald. In open toestand wordt de ingang door een interne schakeling met de 5 Volt voedingspanning hoog gemaakt. Hoog wil zeggen: boven 2,3 Volt, want dit is de spanning waarboven de CMOS-IC HEF4049 een nul niet meer herkent. Omdat echter dit IC tevens invertteert (dit is omkering), "leest" de computer nu een 0 uit. Doorverbinden met de aarde brengt een ingang op nul, maar door de inversie door de IC leest de computer nu een 1 voor de betreffende bit. Het aanleggen van een spanning op een ingang boven 2,3 Volt levert weer een uitlezing van een 0 op het betreffende bit. De spanning mag echter niet meer dan 15 Volt bedragen.

```

5 REM *****
10 REM * Naam programma: DA-TEMP.BAS *
15 REM * Aanmaakdatum : 8-4-1990 *
20 REM * Computer : Philips NMS 9116 *
25 REM * Hardware : Komeet-systeem met ADC8 en NTC. *
30 REM * Beschrijving : Het programma meet de temperatuur met behulp van *
35 REM * de ADC8 en maakt van de metingen een grafiek op *
40 REM * screen 2. *
45 REM * *
50 REM * *
55 REM * *
85 REM * Opmerkingen : ADC8-adres 223. *
90 REM *****
100 KEY OFF:SCREEN 2:CLS:X=40:TIMER ON:REM ***** Naar grafisch scherm
110 GOSUB 290:REM ***** Aanmaak eerste scherm met kop
120 MEET=TIMER+.1
130 X$=INKEY$:IF X$="E" OR X$="e" THEN 210:REM ***** E of e van einde om te stop
pen
140 IF TIMER<MEET THEN 130:REM ***** Na iedere .1 seconde wordt gemeten
145 B%=0:FOR I=1 TO 10:REM ***** Iedere meting wordt in 10-voud gedaan
150 OUT 784,255:OUT 785,223:OUT 784,252:A$=INP(784):REM ***** Meting
155 B%=B%+A$:NEXT:REM ***** B% bevat de som van 10 metingen
160 VIN=B%/2560*2.45:REM ***** Omzetting van gemeten schaaldelen in Volts
165 R1=50/VIN-10:REM ***** Berekening van R1 als R2 10k bedraagt
170 TEMP%=20-50*LOG(R1/100)/LOG(10):REM ***** Berekening van temperatuur
175 Y%=180-TEMP%*2:X=X+1:IF X>639 THEN 100:REM ***** Inschaling voor grafiek
180 IF X=41 THEN PSET(X,Y%):XO=X:YO=Y%:XOO=X:REM ***** Eerste punt grafiek
190 LINE(X,Y%)-(XO,YO):XO=X:YO=Y%:REM ***** Volgende punten grafiek
200 GOTO 120:REM ***** Naar volgende meting
210 SCREEN 0:END:REM stoppen van het programma
280 REM *****
290 PRINT:PRINT" METING VAN DE TEMPERATUUR"
300 REM ***** Aanmaken van een nieuwe grafiekblad
310 LINE (40,180)-(639,40),B:REM ***** Omlijsting
320 LINE (30,130)-(40,130):LOCATE 16,1:PRINT"25"
330 LINE (30,80)-(40,80):LOCATE 10,1:PRINT"50"
340 LOCATE 4,1:PRINT"Graden"
350 RETURN
  
```

```

5 REM *****
10 REM * Naam programma: IO-TEMP.BAS *
15 REM * Aanmaakdatum : 8-4-1990 *
20 REM * Computer : PHILIPS NMS 9100 *
25 REM * Hardware : Komeet-systeem met ADC8 en NTC-weerstand *
30 REM * Beschrijving : Het programma meet de temperatuur van een *
35 REM * NTC-weerstand door de weerstandswaarde te berekenen*
40 REM * uit de tijd nodig om een op 5 Volt opgeladen *
45 REM * condensator tot de helft te ontladen via de *
50 REM * NTC-weerstand en uit de weerstand de temperatuur *
55 REM * te berekenen. *
85 REM * Opmerkingen : DIGOUT-adres 223, INPUT-adres 1 *
90 REM *****
100 OUT 784,0:OUT 785,239:OUT 784,1:OUT 785,0:REM START OPLADEN CONDENSATOR
110 START=TIMER
120 EIND=TIMER:VERSCHILTIDJ=EIND-START:IF VERSCHILTIDJ<60 THEN 120
130 REM NA 60 SEC IS DE CONDENSATOR VAN 100 microF VOLLEDIG OPGELADEN
140 REM *****
200 START=TIMER:OUT 784,0:OUT 785,239:OUT 785,1
210 CLS:LOCATE 10,10:X=INP(784):PRINT X:IF X=1 THEN 300
220 GOTO 210
230 REM DE POORT VAN DE DIGOUT WORDT AFGEZET EN VAN DE INPUT-print WORDT
240 REM POORT 1 GEMETEN. BIJ 2,5 VOLT WORDT DEZE VAN 0 OP 1 GEZET. DAN
250 REM WORDT VERVOLGD OP REGEL 300.
260 REM *****
300 EIND=TIMER:VERSCHILTIDJ=EIND-START
310 X=LOG(VERSCHILTIDJ/6.93)/LOG(10):TEMP=20-X/.02
320 CLS:LOCATE 10,10:PRINT INT(TEMP*10)/10
330 GOTO 100
340 REM DE EINDTIJD WORDT VASTGESTELD EN DE TEMPERATUUR BEREKEND EN GEPRINT.
350 REM DE VOLGENDE MEETCYCLUS WORDT GESTART.
360 REM *****
  
```